Supplement

Power Computations for Intervention Analysis

The result of Tiao et al. (1990) is shown to be a special case of our results in Table 1 for a ramp intervention with AR(1) noise.

Tiao et a. (1990, Appendix A), *p.20,515*

Let $\mathcal{R}_t^{(T)}$ denote the ramp intervention defined in Tiao (1990) so we have $\mathcal{R}_t^{(T)} = R_t^{(T+1)}/12$. Then assuming $T = 0$ the following expression was derived for $\sigma_{\hat{\omega}}$,

$\sigma_{\hat{\omega}}^{(\text{Tiao})} = \sigma\, g(T, \phi)$,

where $\sigma = \sigma_a^2/(1-\phi^2)$ and $g(T, \phi) = 24\,(3\,(1+\phi)/((1-\phi)\,T(T^2-1)))^{1/2}$

In *Mathematica* this expression can be defined as:

```
24 (3 (1 + ϕ) / ((1 - ϕ) T (T^2 - 1)))^(1/2) // Simplify
```

$$24\sqrt{3}\,\sqrt{-\frac{1+\phi}{(-T+T^3)\,(-1+\phi)}}$$

```
gTiao = %
```

$$24\sqrt{3}\,\sqrt{-\frac{1+\phi}{(-T+T^3)\,(-1+\phi)}}$$

Comparison With Our Result

Let $\sigma_{\hat{\omega}}(T, \phi)$ denote our estimate. Then taking into account $\mathcal{R}_t^{(T)} = R_t^{(T+1)}/12$. We obtain, $\mathfrak{g}(T, \phi) = \sqrt{(1-\phi^2)/12}\;\sigma_{\hat{\omega}}(T+1, \phi)$ as the corresponding value in our notation. In order to compare with our result we must assume that the mean is known.

■ Definition of *Mathematica* Function

The results in our Table 1 may be written as:

```
InformationMatrixAR1Ramp[n_, T_, ϕ_] :=
 With[{κ = ϕ - 1,
   b12 = (1 - ϕ) (n + 1 - T) (n + 2 - T - ϕ (n - T)) / 2,
   b22 = 1/6 (1 + n - T) (6 + 7 n + 2 n^2 - 7 T - 4 n T + 2 T^2 - 8 n ϕ - 4 n^2 ϕ + 8 T ϕ + 8 n T ϕ -
       4 T^2 ϕ + n ϕ^2 + 2 n^2 ϕ^2 - T ϕ^2 - 4 n T ϕ^2 + 2 T^2 ϕ^2)}, {{n κ^2, b12}, {b12, b22}}
 ]
```

■ Special Case

The special case corresponds to mean known so,

```
g[T_, ϕ_] :=
 (Sqrt[Inverse[InformationMatrixAR1Ramp[T, 1, ϕ]] // Flatten // Last]) * Sqrt[1 - ϕ^2] * 12

g[T, ϕ] // FullSimplify
```

$$24 \sqrt{\frac{1}{(-T+T^3)\ (-1+\phi)^2}}\ \sqrt{3-3\,\phi^2}$$

■ Proof of Equivalence

Both expressions are positive, so equality of the squared result establishes the equivalence.

```
gTiao^2 - g[T, ϕ]^2 // FullSimplify
```

```
0
```